Introduction to GAMS for economic equilibrium problems\textsuperscript{1}

James R. Markusen
University of Colorado, Boulder

Part A: introduction via a simple supply and demand model: two inequalities in two unknowns

Part B: towards general equilibrium: one good, one factor, one household

This version: July 4, 2005

\textsuperscript{1}This material is the copyrighted intellectual property of James Markusen. Please feel free to use it for individual self-study. But for any use involving fees or changes, however indirect (e.g., course fees), written permission must be obtained from the author.
1. **Formulating an economic equilibrium problem**

The purpose of this section is to introduce you to the use of GAMS for economic equilibrium problems. I have tried to write this so that you can read through it with no GAMS experience, but that might prove awkward at times. You can also read through relevant sections of the GAMS users manual to familiarize yourself with some of the basic GAMS syntax. There are some problems with that manual, particularly for economic modeling however. Perhaps you can pick up a lot of the GAMS syntax from these notes, which focus much more on the types of problems of interest for general-equilibrium modelers.

I teach by using examples; indeed, I originally started this to teach myself. This first example is a simple supply and demand model of a single market, a partial equilibrium model. There are two equations, supply and demand, and two variables, price and quantity. Many types of software can solve such a model of two equations in two unknowns, assuming that it has an interior solution (both price and quantity are positive at the solution). But we will use a more general formulation which is almost exclusively a feature of GAMS that allows for “corner” solutions in which one or more variables is zero. This is an extremely important and powerful feature for use in modeling.

Our approach is to formulate the equations and variables as a complementarity problem. This involves (a) associating each equation with a particular variable, called the complementary variable. (b) if the variables are restricted to be non-negative (prices and quantities), then the equations are written as weak inequalities. If the equation holds as an equality in equilibrium, then the complementary variable is generally strictly positive. If the equation holds as a strict inequality in equilibrium, the complementary variable is zero.
Consider first supply of good X with price P. The “supply curve” in simple competitive models is just the firm’s (or industry’s) marginal cost curve. The supply curve exploits the firm’s optimization decision, which involves equating price with marginal cost: \( P = MC \). However, the good might not be produced since marginal cost might exceed price in equilibrium. The correct condition for equilibrium is:

\[ MC \geq P \quad \text{with the complementarity condition that } X \geq 0 \]

If the equation hold with equality, X is (generally) strictly positive; if it holds with strict inequality, then X is 0. Note that the price equation is complementary with a quantity variable.

Let the cost function for X be given by \( COST = aX + \frac{b}{2} X^2 \). Marginal cost is given by \( MC = a + bX \). The equation for supply, which is associated with the quantity variable X is then:

\[ a + bX \geq P \quad \text{complementary with } X \geq 0. \]

Now consider demand. Optimizing consumer utility for a given income and prices will yield a demand function of the form \( X = D(P, I) \) where I is income. But we will also need to treat the demand side of the model in a complementary fashion, since some goods might be free \((P = 0)\) in equilibrium. The complementarity formulation for the demand side of the model is thus:

\[ X \geq D(P, I) \quad \text{with the complementary condition that } P \geq 0. \]

If the weak inequality holds as an equation, P with be strictly positive; if it holds as a strict inequality in equilibrium, X is free and \( P = 0 \). Note that the quantity equation is complementary with a price variable.

In textbook partial-equilibrium models with income suppressed, we just use the slope and intercept parameters for this function: \( X = c + dP \) where \( c > 0, d < 0 \). Our inequality is then:

\[ X \geq c + dP \quad \text{complementary with } P \geq 0. \]

We now have our model: two equations in two unknowns, each associated with the correct complementary variable. Let’s now see how this is coded and solved in GAMS.

2. Coding an economic equilibrium problem in GAMS

First, comment statements can be used at the beginning of the code, preceded with a *.

Actual code is shown in courier font.
Now we can begin a series of declaration and assignment statements. Here is a list of parameters. If there is text after the parameter with no comma, GAMS does not interpret this text. GAMS saves the text and uses it when parameters are displayed. A semi-colon must be used at the end of the list of parameters.

PARAMETERS
A         intercept of supply on the P axis (MC at Q = 0)
B         change in MC in response to Q, this is dP over dQ
C         intercept of demand on the Q axis (demand at P = 0)
D         response of demand to changes in price, dQ over dP
TAX       a tax rate used later for experiments;

Parameters must be assigned values before the model is solved, but the assignments need not go here, they could appear later in the model. Each assignment statement is separate and so each must end in a semi-colon. These parameter values are referred to as Case 1 below.

A = 2;
C = 6;
B = 1;
D = -1;
TAX = 0;

Now we declare a list of variables. They must be restricted to be positive to make any economic sense, so declaring them as “positive variables” tells GAMS to set lower bounds of zero on these variables. The list must end with a semi-colon. Again, it is optional to add descriptive text after the set labels.

POSITIVE VARIABLES
P
X;

Now we similarly declare a list of equations. We can name them anything we want provided it is a name not otherwise in use or, of course, a keyword. The list ends in a semi-colon.

EQUATIONS
DEMAND
SUPPLY;

Now we specify our equations. The format is to give the equation name followed by two periods (full stops). Then after a space, the equation is written out in the following way, with =G= GAMS code for “greater than or equal to”. 

* M0A.GMS  introductory model using MCP
* simple supply and demand model (partial equilibrium)
SUPPLY.  A + B*X  =G=  P;
DEMAND.  X  =G=  C + D*P;

Next we need to declare a model: a set of equations and unknowns (above, we could have declared more equations and variables than we are actually going to use in the model, so we need to tell GAMS what exactly is the n by n system that we want to solve). The format is the keyword model, followed by a model name of your choosing. Here we use equil for our name. Next comes a “/” followed by a list of the equation names: each equation ends with a period followed by the name of the complementary variable. Each equation.variable is followed by a comma before the next one. The model declaration ends with another “/” and a semi-colon.

MODEL EQUIL /SUPPLY.X, DEMAND.P/;

Finally, we need to tell GAMS to solve the model and what software is needed (GAMS does many other types of problems, such as optimization). Here is the correct statement (MCP stands for mixed complementarity program).

SOLVE EQUIL USING MCP;

This example uses parameter values which generate an “interior solution”, meaning that both X and P are strictly positive. It is shown in the top panel of Figure 1 and labeled “Case 1”. But other outcomes are also possible, and these are both important and common in general equilibrium modeling.

Case 2: It is often the case that a good, or rather a particular way to produce or obtain a good (e.g., via imports) is too expensive relative to some alternative so that this production or trade activity is not used in equilibrium. We sometimes say that this activity is “slack” in equilibrium.

This situation is shown in the second panel of Figure 1: in partial equilibrium, the supply curve lies entirely above the demand curve. The correct solution to the model is that X = 0, and P = 6, the latter being the “choke” price at which demand is zero. To see this, refer back to the equations, and note that the supply equation is slack, but that the demand equation must hold with equality at P = 6. Note especially that, if we didn’t impose the inequality restrictions on P and Q, the model solution would yield a nonsensical solution of X < 0. This case can be generated by the following parameter value of the supply (marginal cost) function. There is no need to respecify the model: just change the value of A and repeat the solve statement.

A = 7;
SOLVE EQUIL USING MCP;

Case 3: The final possibility is that a good or factor of production may be so plentiful that it commands a zero price in equilibrium. This actually happens, for example, in general equilibrium models when activities have fixed coefficients, such as requiring labor and capital in
fixed proportions. Often, this can lead to an equilibrium in which one factor is in excess supply commanding a zero price.

This situation is shown in the third or bottom panel of Figure 1: in the positive orthant, the supply curve lies entirely to the right of the demand curve. The correct solution to this model is that the supply equation holds with equality at X = 7, and the demand equation is slack, P = 0. *Note especially that, if we didn’t impose the inequality restrictions on P and Q, the model solution would yield a nonsensical solution of P < 0.* (Nonsensical if free disposal is assumed; there are cases where it makes sense that the producer pays to have X taken away.) This case can be generated by the following parameter value of the supply (marginal cost) function.

\[ A = -7; \]
\[ \text{SOLVE EQUIL USING MCP;} \]

3. **Counterfactual (comparative statics) experiments**

We don’t want to get into too much detail in this section, but one final thing that can be easily introduced is to show how comparative statics experiments are done, and the use of parameters to extract information after the solve statement.

In the following example, we use a parameter called “TAX”, an ad valorem tax on the input (marginal cost) to production. To do so, we declare and specify an additional equation “SUPPLY2” and model “EQUIL2” (GAMS does not allow us to simply rewrite the equation or use the old model name once we name a new equation). \(^4\) Three parameters are declared and used to extract information after we solve the model.

\begin{verbatim}
PARAMETERS
    CONSPRICE   consumer price
    PRODPRICE   producer price (equal to marginal cost)
    TAXREV      tax revenue (note tax base is producer price);

EQUATIONS
    SUPPLY2;
    SUPPLY2..   \((A + B*X) \times (1+\text{TAX}) =G= P;\)

MODEL EQUIL2 /SUPPLY2.X, DEMAND.P/;
\end{verbatim}

\(^4\)We could of course have put the tax parameter in the model in the first place and given it a zero value. But we are trying to introduce one thing at a time.
Before we solve the model, we need to first return the supply intercept to its original value: parameter changes are kept as permanent unless “undone” in the code. Then we resolve the base case (strictly speaking, this is not necessary, it is just recomputing our first case.).

\[ A = 2; \]
\[ \text{TAX} = 0; \]
\[ \text{SOLVE EQUIL2 USING MCP;} \]

Now set the tax at 25% and solve the model.

\[ \text{TAX} = 0.25; \]
\[ \text{SOLVE EQUIL2 USING MCP;} \]

After we solve the model, it is common to extract output using GAMS code. Note that with the tax specified on inputs, the price being solved for is the consumer price (price paid by the consumer) not the producer price (price received by the producer). Producer price is the same as marginal cost.

GAMS stores three values for variables: its current value, its upper bound (infinity in our case) and its lower bound (zero in our case). When the modeler wants the value of a variable, you have to use the notation \( \text{NAME.L} \) where the \( L \) stands for level. Here is the correct notation for extracting the consumer price, producer price, and total tax revenue. GAMS does not automatically print out the values of parameters in the solution, so you have to ask GAMS to \text{DISPLAY} these values.

\[ \text{CONSPRICE} = \text{P.L}; \]
\[ \text{PRODPRICE} = \text{P.L}/(1+\text{TAX}); \]
\[ \text{TAXREV} = \text{PRODPRICE*TAX*X.L}; \]
\[ \text{DISPLAY CONSPRICE, PRODPRICE, TAXREV;} \]

4. **Reading the output**

Again, you need to consult the GAMS users’ manual to see how to run the model, and to read and interpret output once you run the model. There is a lot of “stuff” in GAMS listing (output) files, which are stored as \textit{FILENAME.LST} after you run the model \( (LST \) is for “listing file”). Here are just the relevant parts of our model runs. But first, for any/all solve statement, check to make sure that you see the following.
SOLVE SUMMARY
MODEL EQUIL
TYPE MCP
SOLVER MILES FROM LINE 50

**** SOLVER STATUS 1 NORMAL COMPLETION
**** MODEL STATUS 1 OPTIMAL

Case 1: The LEVEL is the solution value of the variables. MARGINAL indicates the degree to which the equation corresponding to the variable is out of equality. For \( P \) (price), the equation is DEMAND and the value of the marginal is supply minus demand. For \( X \) (quantity), the equation is SUPPLY and the value of the marginal is the excess of marginal cost over price. Variables that have positive values in the solution should have zero marginals. Variables that have zero values in the solution should have positive marginals.

<table>
<thead>
<tr>
<th>LOWER</th>
<th>LEVEL</th>
<th>UPPER</th>
<th>MARGINAL</th>
</tr>
</thead>
<tbody>
<tr>
<td>4.000</td>
<td>4.000</td>
<td>+INF</td>
<td>.</td>
</tr>
<tr>
<td>2.000</td>
<td>2.000</td>
<td>+INF</td>
<td>.</td>
</tr>
</tbody>
</table>

Case 2: This is the zero-output case. The price equation holds, but the quantity equation is slack. The marginal of 1.0 indicates that, at the solution, marginal cost exceed price by 1.0.

<table>
<thead>
<tr>
<th>LOWER</th>
<th>LEVEL</th>
<th>UPPER</th>
<th>MARGINAL</th>
</tr>
</thead>
<tbody>
<tr>
<td>6.000</td>
<td>6.000</td>
<td>+INF</td>
<td>.</td>
</tr>
<tr>
<td>.</td>
<td>.</td>
<td>+INF</td>
<td>1.000</td>
</tr>
</tbody>
</table>

Case 3: This is the free-good case. Now the price equation is slack, and the marginal of 1.0 indicates that, at the solution, supply exceeds demand by 1.0.

<table>
<thead>
<tr>
<th>LOWER</th>
<th>LEVEL</th>
<th>UPPER</th>
<th>MARGINAL</th>
</tr>
</thead>
<tbody>
<tr>
<td>.</td>
<td>.</td>
<td>+INF</td>
<td>1.000</td>
</tr>
<tr>
<td>7.000</td>
<td>7.000</td>
<td>+INF</td>
<td>.</td>
</tr>
</tbody>
</table>

Tax example: Here is the tax example (after the supply intercept is set back at 2.0). Note the “tax incidence” is split between the producer and consumer: the initial price was 4.0.

<table>
<thead>
<tr>
<th>LOWER</th>
<th>LEVEL</th>
<th>UPPER</th>
<th>MARGINAL</th>
</tr>
</thead>
<tbody>
<tr>
<td>4.444</td>
<td>4.444</td>
<td>+INF</td>
<td>.</td>
</tr>
<tr>
<td>1.556</td>
<td>1.556</td>
<td>+INF</td>
<td>.</td>
</tr>
</tbody>
</table>

82 PARAMETER CONSPRICE = 4.444 consumer price
82 PARAMETER PRODPRICE = 3.556 producer price
82 PARAMETER TAXREV = 1.383 tax revenue
5. The entire model, uninterrupted by talk

* M0.GMS introductory model using MCP
* simple supply and demand model (partial equilibrium)

PARAMETERS
A         intercept of supply on the P axis (MC at Q = 0)
B         change in MC in response to Q, this is dP over dQ
C         intercept of demand on the Q axis (demand at P = 0)
D         response of demand to changes in price, dQ over dP
TAX       a tax rate used later for experiments;

A = 2;
C = 6;
B = 1;
D = -1;

POSITIVE VARIABLES
  P
  X;

EQUATIONS
DEMAND
  SUPPLY;

  SUPPLY..  A + B*X  =G=  P;

  DEMAND..  X  =G=  C + D*P;

MODEL EQUIL /SUPPLY.X, DEMAND.P/;

SOLVE EQUIL USING MCP;

A = 7;
SOLVE EQUIL USING MCP;

A = -7;
SOLVE EQUIL USING MCP;

PARAMETERS
  CONSPRICE   consumer price
  PRODPRICE   producer price (equal to marginal cost)
  TAXREV      tax revenue (note tax base is producer price);
EQUATIONS
SUPPLY2;

SUPPLY2.. (A + B*X)*(1+TAX) =G= P;

MODEL EQUIL2 /SUPPLY2.X, DEMAND.P/;

A = 2;
TAX = 0;
SOLVE EQUIL2 USING MCP;

TAX = 0.25;
SOLVE EQUIL2 USING MCP;

CONSPRICE = P.L;
PRODPRICE = P.L/(1+TAX);
TAXREV = PRODPRICE*TAX*X.L;
DISPLAY CONSPRICE, PRODPRICE, TAXREV;

Exercise:

(1) Try associating the wrong variables with the equations in the model statement. Run the model and see what happens.

(2) Consumer surplus is (geometrically) the area of the triangle under the demand curve above consumer price. Define a new parameter CONSURP and assign it the right value. Display its value with and without the tax.

Hints: the intercept of the demand curve on the price axis is -C/D.

The height of the triangle is (-C/D - P.L), and its base is X.L.

The area is (-C/D - P.L)*X.L/2.
Figure 1: Three outcomes of partial equilibrium example

Case 1: interior solution

Figure 1: Three outcomes of partial equilibrium example

Case 2: X is too expensive, not produce in equilibrium (X = 0)

Case 3: excess supply, X is a free good in equilibrium (P = 0)
1. **Toward General Equilibrium**

In the previous section we used a partial-equilibrium model to introduce complementarity and the GAMS solver. That model does not specify what is used to produce X (labor?) and it does not specify what the consumer uses to purchase X (income, derived how?).

In this section, we want to take a step toward general equilibrium, completing the relationship between producers and consumers to form a complete economic system. The model is simple in economic terms, but it is important for building our understanding of how to translate an economic model into a computable, applied general-equilibrium model. We will argue at the end of the section that the way we represent this simple economy is indeed very general and carries over to far more complex economies.

Our model has one household and one firm. The household has a fixed stock of labor (L). We make the further assumption that the household derives no utility from leisure, and so will always supply the entire stock of labor to the market if the wage rate (W) is positive. There is a single good (X) produced from the single input labor. The firm buys labor services and sells X to the household at price P. The household receives income from selling labor services and uses it all to buy X.

Our model is shown schematically in Figure 2. There are four interactions between the household and the firm. In physical units, the household supplies labor to the firm and receives X output. In payments, the household receives labor income from the firm and pays the firm for X. As noted in the Figure, general equilibrium requires five conditions:

1. the consumer optimizes
2. the producer optimizes
3. labor demand equals labor supply (the labor market clears)
4. X demand equals X supply (the goods market clears)
5. consumer income equals expenditure (income-expenditure balance)

In our model, the consumer has no alternative use for labor and so optimizes by supplying it all. The consumer prefers more X and so the demand for X is going to be given by $X = WL/P$: it is optimal to spend all income on X.

The production function for X will be given as $X = \alpha L$, where $\alpha$ is the marginal product of labor in X production. We assume competition and free entry, so that the firm is force to price at marginal cost. Marginal cost, the cost of producing one more unit of X, is given by $W/\alpha$.
(1/α is the amount of labor needed for one unit of X).

Our model has only two parameters, total labor supply (LBAR) and productivity α (ALPHA).

PARAMETERS
LBAR labor supply (fixed and inelastic)
ALPHA productivity parameter  X = ALPHA*L;

LBAR = 100;
ALPHA = 2;

There will be four positive variables: price (P), quantity of X (X), the wage rate (W), and consumer income (INCOME).

POSITIVE VARIABLES
P price of X
X quantity of X
W wage rate
INCOME income from labor supply;

There are four equations. One important feature of general-equilibrium models with constant returns to scale as assumed here, is that there is really no “supply” function in the usual partial-equilibrium sense: at given factor prices, the “supply” curve (the marginal cost curve) is horizontal. Supply is perfectly elastic at a fixed price P = MC, the latter being a constant at constant factor prices.

Instead of using the concept of supply, we use the pricing equation of firms directly, which requires that marginal cost is greater than or equal to price, with output being the complementary variable to this inequality. With constant returns, marginal cost is also average cost, and so this equation can also be thought of as a free-entry, zero-profits condition. Accordingly, we refer to this equation as ZPROFIT (zero profits).

Then we require two market-clearing conditions. First, supply of X must be greater than or equal to demand with the complementary variable being P, the price of X. This is referred to as CMKTCLEAR (commodity-market clearing). Second, the supply of labor must be greater than or equal to its demand, with the complementary variable being W, the wage rate (price of labor). This is labeled LMKTCLEAR (labor-market clearing). Finally, we require income balance: consumer expenditure equals labor income. This is labeled CONSINCOME.
The equations are very simple. Marginal cost, equal average cost, is just the wage rate
divided by $\alpha$, the productivity parameter as discussed above. Commodity market clearing
requires that $X$ in equilibrium is greater than or equal to demand, which is just income divided
by $P$, the price of $X$. Labor-market clearing requires that supply, $\text{LBAR}$, is greater than or equal
to labor demand, given by $X$ divided by $X/\alpha$. Income spent on $X$ must equal wage income,$W$ times $\text{LBAR}$ ($W*\text{LBAR}$).

$$ZPROFIT.. \quad W/\alpha = G= P;$$

$$CMKTCLEAR.. \quad X = G= \text{INCOME}/P;$$

$$LMKTCLEAR.. \quad \text{LBAR} = G= X/\alpha;$$

$$CONSINCOME.. \quad \text{INCOME} = G= W*\text{LBAR};$$

We name the model “GE” for general equilibrium, and associate each equation in the model with
its complementary variable.

MODEL GE /ZPROFIT.X, CMKTCLEAR.P, LMKTCLEAR.W,
CONSINCOME.INCOME/;

Now we introduce a new feature, namely, setting starting values of variables. This helps
the solver find the solution, and can be quite important in complex problems. The notation for
setting an initial value of a variable is the NAME.L notation we used earlier.

* set some starting values

$$P.L = 1;$$
$$W.L = 1;$$
$$X.L = 200;$$
$$\text{INCOME}.L = 100;$$

One final issue that is familiar to students of economics, is that there is an indeterminacy
of the “price level” in this type of problem. If $P = W = 1$ and $\text{INCOME} = 100$ are solutions to
this model, so are any proportional multiples of these values such as $P = W = 2$, $\text{INCOME} = 200$. 
More formally, there are really only three independent equations in this model, the fourth is automatically satisfied if three hold. This is known in economics as “Walras’ Law”.

The solution to this problem is to fix one price, termed the “numeraire”. Then all prices are measured relative or in terms of this numeraire. Suppose that we choose the wage rate to be the numeraire. The notation is “W.FX” where .FX stands for “fix”. It is important to understand that W.FX is not the same as W.L. The former holds W fixed throughout the remainder of the program, whereas W.L is just setting an initial value that will be changed by the solver (unless it happens to be the correct equilibrium value of course). When a variable is fixed, GAMS automatically drops the complementary equation from the model.

* choose a numeraire

\[ W.FX = 1; \]

Now solve the model. As a counter-factual experiment, double labor productivity and resolve.

```
SOLVE GE USING MCP;
* double labor productivity
ALPHA = 4;
SOLVE GE USING MCP;
```

Here is the solution to the first solve statement. Notice that W.FX shows up as fixing both the upper and lower bounds for W.

```
                  LOWER     LEVEL     UPPER    MARGINAL
   ---- VAR P   .        0.500     +INF       .
   ---- VAR X   .      200.000     +INF       .
   ---- VAR W   1.000     1.000     1.000      EPS
   ---- VAR INCOME .      100.000     +INF       .
```

Here is the solution to the second solve statement. Doubling of labor productivity allows twice as much X to be produced from the fixed supply of labor. With the wage rate fixed at \( W = 1 \), the equilibrium price of X falls in half: twice as much X can be purchased from the income from a unit of labor.
In order to appreciate correctly the role of the numeraire, suppose instead we wanted to choose X as numeraire. The first thing that we have to do is “unfix” W. This actually requires two separate statements, since .FX involves fixing both the lower and upper bounds of W. Then we set P as numeraire.

```plaintext
W.UP = +INF;
W.LO = 0;
P.FX = 1;
SOLVE GE USING MCP;

* double labor productivity
ALPHA = 4;
SOLVE GE USING MCP;
```

Here are the solutions for the two runs of our model.

<table>
<thead>
<tr>
<th>LOWER</th>
<th>LEVEL</th>
<th>UPPER</th>
<th>MARGINAL</th>
</tr>
</thead>
<tbody>
<tr>
<td>---- VAR P</td>
<td>1.000</td>
<td>1.000</td>
<td>1.000</td>
</tr>
<tr>
<td>---- VAR X</td>
<td>.</td>
<td>200.000</td>
<td>+INF</td>
</tr>
<tr>
<td>---- VAR W</td>
<td>.</td>
<td>2.000</td>
<td>+INF</td>
</tr>
<tr>
<td>---- VAR INCOME</td>
<td>.</td>
<td>200.000</td>
<td>+INF</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>LOWER</th>
<th>LEVEL</th>
<th>UPPER</th>
<th>MARGINAL</th>
</tr>
</thead>
<tbody>
<tr>
<td>---- VAR P</td>
<td>1.000</td>
<td>1.000</td>
<td>1.000</td>
</tr>
<tr>
<td>---- VAR X</td>
<td>.</td>
<td>400.000</td>
<td>+INF</td>
</tr>
<tr>
<td>---- VAR W</td>
<td>.</td>
<td>4.000</td>
<td>+INF</td>
</tr>
<tr>
<td>---- VAR INCOME</td>
<td>.</td>
<td>400.000</td>
<td>+INF</td>
</tr>
</tbody>
</table>

In comparing the two alternative choices of numeraire, note that the choice does not affect the only “real” variable of welfare significance, the output of X. The variables affected are “nominal” variables. But with no money or nominal assets, the price level is of no significance. Alternatively, only relative prices matter and the “real wage”, W/P is the same.
under either choice of numeraire.

2. **A general formulation of general-equilibrium models**

The way that we formulated this very simple model is just a special case of a very general methodology that proves to be useful and robust in much more complicated situations. The purpose of this section is to discuss this general formulation.

First, general-equilibrium models consist of activities that transform some goods and factors into others. These include outputs of goods from inputs, trade activities that transform domestic into foreign goods and vice versa, activities that transform leisure into labor supply, and activities that transform goods into utility (welfare).

Activities are most usefully represented by their dual, or cost-functions. The conditions for equilibrium are then that marginal cost for each activity is greater than or equal to price, with the complementary variable being the equilibrium quantity or “level” that activity. A quantity variable is complementary to a price equation. With competition and constant returns to scale, these conditions are also referred to as *zero profit conditions*.¹

Second, general-equilibrium models consist of market clearing conditions. A commodity is a general term that includes goods, factor of production, and even utility. Thus X and labor are both commodities in our example. Activities transform some commodities into other commodities. Market clearing conditions require that the supply of a commodity is greater than or equal to its demand in equilibrium, with the complementary variable being the price of that commodity. A price variable is complementary with a quantity equation.

Finally, there are income-balance equations for each “agent” in a model. Agents are generally household, but often include a government sector, or the owner of a firm in models with imperfect competition and pure profits. Expenditure (Exp) equals income for each agent.

Let i subscript activities, also referred to as production sectors. Let j subscript commodities, and let k subscript agents or households. Our general formulation is then:

---

¹This is not true with imperfect competition or increasing returns to scale: the producer optimization condition need not imply zero profits. For now, we will stick with competition and constant returns, under which producer optimization is equivalent to zero profits.
The term “price” (P) here could be a bit misleading insofar as one activity could produce multiple commodity outputs. If the outputs are in fixed proportion, for example, then the output “price” of the activity is actually the sum of the prices of the joint commodity outputs.

Alternatively, a price index can be fixed as numeraire, such as prices summing to 1. A common choice is that the real consumer price index, the cost of buying one unit of utility, can be held constant. This is an economically convenient choice, because then all prices (wages, rents, goods) are reported as “real” prices.

<table>
<thead>
<tr>
<th>Inequality</th>
<th>Complementary var.</th>
<th>No. of eq. unknowns</th>
</tr>
</thead>
<tbody>
<tr>
<td>(1) Zero profits²</td>
<td>MCₖ ≥ Pₖ</td>
<td>Activity (quantity) level i i</td>
</tr>
<tr>
<td>(2) Market clearing</td>
<td>S_j ≥ D_j</td>
<td>Price of commodity j j</td>
</tr>
<tr>
<td>(3) Income balance</td>
<td>Exp_k ≥ Income_k</td>
<td>Income k k</td>
</tr>
</tbody>
</table>

The general-equilibrium system then consists of i+j+k inequalities in i+j+k unknowns: i activity levels, prices of j commodities, and the incomes/expenditures of k agents. As per our earlier discussion, the price level is indeterminate and one price is fixed as a numeraire.³

3. The entire model, uninterrupted by talk

PARAMETERS
LBAR    labor supply (fixed and inelastic)
ALPHA   productivity parameter X = ALPHA*L;

LBAR = 100;
ALPHA = 2;

POSITIVE VARIABLES
P         price of X
X         quantity of X
W         wage rate
INCOME    income from labor supply;

EQUATIONS
ZPROFIT      zeroprofits in X production
CMKTCLEAR    commodity (X) market clearing
LMKTCLEAR    labor market clearing
CONSINCOME   consumer income;

²The term “price” (P) here could be a bit misleading insofar as one activity could produce multiple commodity outputs. If the outputs are in fixed proportion, for example, then the output “price” of the activity is actually the sum of the prices of the joint commodity outputs.

³Alternatively, a price index can be fixed as numeraire, such as prices summing to 1. A common choice is that the real consumer price index, the cost of buying one unit of utility, can be held constant. This is an economically convenient choice, because then all prices (wages, rents, goods) are reported as “real” prices.
ZPROFIT.. W/ALPHA =G= P;
CMKTCLEAR.. X =G= INCOME/P;
LMKTCLEAR.. LBAR =G= X/ALPHA;
CONSINCOME.. INCOME =G= W*LBAR;
MODEL GE /ZPROFIT.X, CMKTCLEAR.P, LMKTCLEAR.W,
CONSINCOME.INCOME/;

* set some starting values
P.L = 1;
W.L = 1;
X.L = 200;
INCOME.L = 100;

* choose a numeraire
W.FX = 1;
OPTION MCP = PATH;
SOLVE GE USING MCP;

* double labor productivity
ALPHA = 4;
SOLVE GE USING MCP;

* change numeraire
W.UP = +INF;
W.LO = 0;
P.FX = 1;
ALPHA = 2;
SOLVE GE USING MCP;

* double labor productivity
ALPHA = 4;
SOLVE GE USING MCP;
Conditions for general equilibrium

1. Firms optimize
2. Consumers optimize
3. Supply equals demand for labor
4. Supply equals demand for good X
5. Income equals expenditure (all labor income is spent on X)